

A partial account of verbal case in Kayardild

Dunstan Brown

0. Introduction

Kayardild is a language of the Tangkic group and is traditionally spoken on the South Wellesley Islands (Bentinck Island, Sweers Island, several smaller islands and sometimes Allen Island) in Australia. Amongst other things, it is famous for its verbal cases. It is described in great detail in Evans (1995). There is an entry for it in the Typological Database on Deponency (Baerman 2006a) under the Feature tree word class//noun//noun/verb, and paradigmatic//case under Domain. Because of verbal case the noun therefore looks morphologically like a verb, marking tense, aspect and mood.

The noun paradigm is illustrated by the noun *mala* 'sea'.

<i>mala</i> - 'sea'	
NOM	mala-a
LOC	mala-ya
ABL	mala-na
PROP	mala-wuru
OBL	mala-ntha
ALL	mala-r
GEN	mala-karra
ASSOC	mala-nurru
ORIG	mala-wan-
PRIV	mala-warri
CONS	mala-ngarrba
INS	mala-nguni
UTIL	mala-marra
VALL	mala-kiiwa-tha
VDAT	mala-maru-tha
VTRANS	mala-marii-ja
VABL	mala-wula-tha (<i>or</i> -wula-a-ja)
VEVIT	mala-waalu-tha (<i>or</i> -waala-i-ja)
VDON	mala-wu-ja
VPURP	mala-jani(i)-ja

(Evans 1995: 125, 165)

The relevant items for our purposes are those cases which begin with V in the table. VDAT, for example, is the verbal dative. The form *-maru-* after the root of the noun licenses the use of TAM marking, in this table the actual marker *-tha/-ja*.

The challenge for a formal, implemented theory of this phenomenon is to allow for the use of verbal cases without permitting just anything to happen. Our fragment illustrates how *malaa* is assigned the ordinary cases, and how the verbal markers license the use of the TAM marking. We additionally try to show how the conjugation difference *-tha/-ja* in the TAM marking can be inherited via the original verb source. We therefore also give the original verbs which some of the verbal cases are based on and show how they can function as basic verbs and also be used in the noun paradigm. The theorems which can be derived from this small fragment show that a verb will have the value ‘undefined’ for the ordinary noun cases and only outputs the actual TAM form. Hence, the theory does not allow for the deponency by saying just anything can happen. We have not included the modal cases in the Network Morphology analysis, as we have aimed to concentrate on the point highlighted on the typological database.

1. Overview of the Network Morphology account

Key distinctions in the account

- There is a morphological hierarchy in the fragment consisting of a MOR_WORD node which makes statements about words in Kayardild in general.
- The node MOR_NOUN inherits from MOR_WORD information about morphotactics and co-occurrence constraints which stop verbs from realizing case (in the absence of information to licenses this) and nouns from realizing TAM. The constraint for nouns is `<mor noun tam> == UNDEFINED`, which makes use of the ordering of attributes in DATR to prohibit the realization of TAM information on nouns in the absence of verbal case. A similar method is used to constrain the appearance of case on verbs.
- In the evaluated theory the TAM information is an extension of verbal case paths. For example: `<mor case verbal all tam act>` is the verbal allative with actual TAM.
- Quoted nodes are used to inherit the TAM verb morphology when there is verbal case. For example: `<mor case verbal dat> == - "Marutha:<mor word>"`. Informally, the quoted node construct sets the verb node *Marutha*, for example, as determining the routes of inheritance for marking the verbal dative. This also means that the verbal dative inherits the actual TAM marking *-tha*, rather than the palatal conjugation’s *-ja*. (We have not included the other TAM marking in the fragment, as we are only illustrating the principle.)

A sketch of the Network Morphology account

Nouns inherit noun morphology from a MOR_NOUN node. Verbs inherit verb morphology from one of two conjugation class nodes (V_DENTAL or V_PALATAL), depending on the

verb. Ultimately these nodes will inherit from the MOR_WORD node, which looks like this:

```
MOR_WORD:
  <> == UNDEFINED
  <mor word> == "<mor "<syn cat>" >"
  <mor tact> == "<root>" "<mor>"
  <mor noun tam> == UNDEFINED
  <mor verb case> == UNDEFINED
  <mor noun> == <mor tact>
  <mor verb> == <mor tact>.
```

If there is no specification of information in noun morphology or the nodes associated with the verb conjugations, then the system defaults to this node to determine the morphology. If MOR_NOUN does not specify the required information, then it inherits from UNDEFINED (i.e. the system does not provide an answer or the queried information is not coherent.). At MOR_NOUN a word requires evaluation of syntactic category. If the word is a noun and the query involves TAM, then the value will be undefined. (This is a kind of co-occurrence constraint.) Similarly, for verbs and case, there will be no value returned other than the undefined one. These constraints will not apply for verbal case, as TAM does not start the noun path, but extends the verbal case path. A good informal way of understanding this is to see it as saying that TAM is not at the root of the noun paradigm, but can be licensed to occur as a further subdivision by other elements.

This paradigmatic structure is clear from the MOR_NOUN node:

```
MOR_NOUN:
  <> == MOR_WORD
  <mor case nom> == -a
  <mor case loc> == -ya
  <mor case abl> == -na
  <mor case prop> == -wuru
  <mor case obl> == -ntha
  <mor case all> == -r
  <mor case gen> == -karra
  <mor case assoc> == -nurru
  <mor case orig> == -wan-
  <mor case priv> == -warri
  <mor case cons> == -ngarrba
  <mor case ins> == -nguni
  <mor case util> == -marra
  <mor case verbal all> == -kiiwa "V_DENTAL:<mor>"
  <mor case verbal dat> == - "Marutha:<mor word>"
  <mor case verbal trans> == - "Mariija:<mor word>"
  <mor case verbal abl> == -wula "V_DENTAL:<mor>"
  <mor case verbal evit> == - "Waalutha:<mor word>"
```

```
<mor case verbal don> == -wu "V_PALATAL:<mor>"
<mor case verbal purp> == - "Janija:<mor word>".
```

As can be seen, the ordinary cases are specified in a straightforward way. The verbal cases have been treated in one of two ways. If the formal correspondence with the original verb source is not so transparent, we have specified the form in the paradigm, as is true for *-kiiwa*, for example. But this also means that we have to say which conjugation the TAM should be taken from (V_DENTAL, in this instance). In contrast, with the verbal dative all of the information is inherited from the lexical entry for the verb *marutha* ‘put’.

Example lexical entries for a noun and a verb

Malaa:

```
<> == NOUN
<root> == mala
<gloss> == sea.
```

Marutha:

```
<> == VERB
<mor> == V_DENTAL
<root> == maru
<gloss> == put.
```

Example Outputs

```
Malaa:<syn cat> = noun.
Malaa:<gloss> = sea.
Malaa:<mor word case nom> = mala -a.
Malaa:<mor word case loc> = mala -ya.
Malaa:<mor word case abl> = mala -na.
Malaa:<mor word case prop> = mala -wuru.
Malaa:<mor word case obl> = mala -ntha.
Malaa:<mor word case all> = mala -r.
Malaa:<mor word case gen> = mala -karra.
Malaa:<mor word case assoc> = mala -nurru.
Malaa:<mor word case orig> = mala -wan-.
Malaa:<mor word case priv> = mala -warri.
Malaa:<mor word case cons> = mala -ngarrba.
Malaa:<mor word case ins> = mala -nguni.
Malaa:<mor word case util> = mala -marra.
Malaa:<mor word case verbal all tam act> = mala -kiiwa -th -a.
Malaa:<mor word case verbal dat tam act> = mala - maru -th -a.
Malaa:<mor word case verbal trans tam act> = mala - marii -j -a.
Malaa:<mor word case verbal abl tam act> = mala -wula -th -a.
Malaa:<mor word case verbal evit tam act> = mala - waalu -th -a.
Malaa:<mor word case verbal don tam act> = mala -wu -j -a.
Malaa:<mor word case verbal purp tam act> = mala - jani -j -a.
Malaa:<mor word tam act> = undefined.
```

2. Conclusion

Our purpose in the small fragment for Kayardild has been to outline how the verbal case marking can involve noun and verb morphology simultaneously. At the same time we wished to show that not just anything is allowed by the theory. We have done this by imposing a structure on the paradigm.

References

Baerman, Matthew. 2006. *The Typological Database on Deponency*.
< <http://www.smg.surrey.ac.uk/deponency> >

Evans, Nicholas. 1995. *A grammar of Kayardild*. Berlin: Mouton de Gruyter.